
Shifting the Pitch of Audio in the 
Frequency Domain

Introduction
This application will shift the pitch of an audio file by scaling its frequency spectrum.

Our simple approach will multiply each frequency by the same amount. For a piano or violin note, the 
harmonics will still be an integer multiple of the pitch with the same balance between amplitudes 
(baring some digital aliasing effects). This preserves the timbre of the sound.

In more detail:

A small WAV file is imported (by default, a violin note played at B5 with vibrato). 
The audio is converted to the frequency domain. Each spectral point corresponds to a 
frequency of 0 Hz to fs/2 Hz in in steps of fs/N (where fs is the sampling rate and N is the 
number of samples). 
A new vector of frequencies is generated by multiplying the original frequencies by 
2n_semitones/12 (where n_semitones is the number of semitones, and can be positive or 
negative)
New spectral data is generated by interpolating the original frequencies and spectral points 
against the new frequencies.
The new spectral data is inverted to the time domain, and converted to audio.

If your original audio is a piano note at F4 (a pitch of 349.2 Hz), you can convert it to an A4 note by 
scaling the frequencies by 4 semitones (a pitch of 349.2 Hz 4/12 = 440.0 Hz).

The pitch-shifted audio requires data that nearly always falls between the bins of the unaltered 
spectral data, but is approximated with linear interpolation. This slightly distorts the pitch-shifted 
audio.

More complex methods of pitch shifting use overlapping slices of audio analyzed with short time 
Fourier transforms (as employed in phase vocoders).

This loudspeaker component is needed to play the audio files

Load Packages
restart:
with(AudioTools):
with(SignalProcessing):
with(ArrayTools):
with(CurveFitting):

Import and Prepare the Original Audio
By default, the application analyzes one of the standard WAV file distributed with Maple - a violin note 
played at B5 with vibrato. 
original_audio := Read( FileTools:-JoinPath( [ kernelopts( datadir
), "audio", "ViolinThreePosVibrato.wav" ] ) )

Sampling rate
fs := attributes( original_audio )[ 1 ];

If the audio has two channels, only analyze the first channel
if numelems( Dimensions( original_audio ) ) = 2 then 
   original_audio := original_audio[ .., 1 ];
end if

Number of samples
N := numelems( original_audio )

Truncate the audio file to an even number of samples (this makes the code simpler, and we won't 

notice a missing sample)
if N mod 2 <> 0 then
   original_audio := original_audio[ .. -2 ]:
   N := N - 1:
end if:

Play the audio file
Play(original_audio)

Shift the Pitch in the Frequency Domain
Transform to the frequency domain
data_fft := Vector[ column ]( FFT( original_audio ) );

Warning, size of Array must be a power of two greater than two, 
using DFT instead and suppressing this warning for the remainder 
of this session

Extract the first and second half of the frequency spectrum. The second half is a mirror image of the 
first half, and is reversed so frequencies increase in the same direction as the first half.
first_half_spectrum  := data_fft[ 1 .. N / 2 ]:
second_half_spectrum := Reverse( data_fft[ N / 2 + 1 .. -1 ] ):

Frequencies in the spectral data of the original audio
freq_orig := Vector( [ seq( i * fs / N, i = 1 .. N / 2) ], 
datatype = float[ 8 ] );

Let's now scale the frequencies up or down by a number of semitones. The first step is to generate a 
new vector of scaled frequencies. 

n_semitones is positive to increase pitch, and negative to decrease pitch (an octave is 12 semitones, 
and is a doubling or halving of the frequency). Increasing the pitch by 2 semitones will change the B5 
violin note to a C#6.
n_semitones := 2:
freq_scaled := freq_orig * 2. ^ ( -n_semitones / 12 );

Now make a vector of new spectral data that contains  interpolated values of the original spectral data.

Specifically, we need to interpolate the value of [freq_orig, data_fft] at 
freq_shift[1] to find the spectral value at the 1st index
freq_shift[2] to find the spectral value at the 2nd index
etc

first_half_spectrum_scaled  :=  ArrayInterpolation( freq_orig, 
simplify( Re~( first_half_spectrum  ) ), freq_scaled, method = 
linear ) + I * ArrayInterpolation( freq_orig, simplify( Im~( 
first_half_spectrum  ) ), freq_scaled, method = linear ):

second_half_spectrum_scaled :=  ArrayInterpolation( freq_orig, 
simplify( Re~( second_half_spectrum ) ), freq_scaled, method = 
linear ) + I * ArrayInterpolation( freq_orig, simplify( Im~( 
second_half_spectrum ) ), freq_scaled, method = linear ):

Combine the two halves of the scaled spectral data
pitch_shifted_spectrum := Concatenate(1, 
first_half_spectrum_scaled, Reverse( second_half_spectrum_scaled )
):

Compare the Original and Pitch-Shifted Audio
Now convert the interpolated spectral data back to the time domain and create an audio file
pitch_shifted_time_domain := Re~( InverseFFT( 
pitch_shifted_spectrum ) ):
pitch_shifted_audio := Create( pitch_shifted_time_domain, rate = 
fs )

Periodograms of the original and pitch-shifted audio
plots:-display( 
 Periodogram( original_audio,      color = "DarkGrey", thickness =
0, legend = "Original Audio" )
,Periodogram( pitch_shifted_audio, color = "DarkGreen",  thickness
= 0, legend = "Pitch Shifted Audio" )
,title = cat( "Shifting the Pitch of an Audio File by ", 
n_semitones, " semitones"), titlefont = [ Arial, 14 ], legendstyle
= [ font = [ Arial ] ]
,size = [800, 400] )

Play the original audio file again
Play( Normalize( original_audio, offset = remove, amplitude = 0.99
) )

Now play the pitch shifted audio
Play( Normalize( pitch_shifted_audio, amplitude = 0.99 ) )

For the violin note analyzed by this application, you can even create a duet of violins by playing the 
original and pitch-shifted audio in either channel of a stereo sound file.
duet := Create( << original_audio, pitch_shifted_audio >> ^ %T, 
rate = fs)

Play( Normalize( duet, amplitude = 0.99 ) )


